

Electrostatic Plasma Simulation

APAM1601y
Columbia University

Introduction

References:

Berkeley Plasma Simulation Group

Prof. Charles "Ned" Birdsall

Plasma Physics Via Computer Simulation, by Birdsall and Langdon, (Institute of Physics, 1991)

This notebook calculates a (very small) "particle-in-cell" (PIC) simulation of a one-dimensional plasma. Electrons and ions interact collectively by way of electrostatic forces. This is a very demanding computation. PIC simulations are usually performed on large supercomputers and require many hours (or days) of computation. Current research using these types of plasma codes are 3D, electromagnetic, and defined with complicated geometries.

Equations

The equations of motion for the i th particle are

$$\partial_t x_i = v_i$$

$$\partial_t v_i = (q_i/m_i) E(x_i, t)$$

The equation for the electric field

$$E(x, t) = -\partial_x \Phi(x, t)$$

$$\partial_{x,x} \Phi = -\rho(x, t) / \epsilon_0$$

where ρ is the total charge density, including both the electrons and the ions.

But, we can speed the calculations if we assume the ions are so heavy that they do not move. This restricts our problems to considering electron dynamics only.

Physics

A plasma is characterized by a *length* scale called the "Debye length", a *time* scale by the (inverse) "plasma frequency", and a *velocity* scale associated with the "thermal velocity".

In MKS units,

$$\omega_p = \sqrt{q_e^2 n / \epsilon_0 m_e}, \text{ proportional to square-root of the density}$$

$$v_{\text{the}} = \sqrt{kT_e / m_e}, \text{ proportional to the square-root of the temperature}$$

$$\lambda_D = v_{\text{the}} / \omega_p, \text{ the ratio of the thermal speed to the plasma frequency}$$

In order for us to properly simulate a plasma, we need a "box" much larger than λ_D and a simulation that lasts much longer than ω_p^{-1} . We also require the numerical time step, Δt , to be much *shorter* than ω_p^{-1} .

Numerics

For simplicity, we'll make the following assignments:

- $\Delta x = \Delta t = 1$ (the spatial and temporal grid points are integers)
- $(q_e/m_e) = -1$ for electrons
- $(q_i/m_i) = 1/25$ for ions. (The ions are *actually* much heavier!)

With $\Delta x = 1$, the total length of our "box" of particles must be $L \gg 1$.

With $\Delta t = 1$ and with $q_e/m_e = 1$, then $\Delta t \omega_p = \sqrt{q_e n / \epsilon_0} \ll 1$. If we have 2000 electrons, then $n = 10^3 / L \sim 1$. Thus, we need to set the parameter $q_e / \epsilon_0 \ll 1$.

With $v_{\text{the}} = \lambda_D / \omega_p$, then $v_{\text{the}} \sim 1$.

Basic Definitions: Heavy stationary ions

```
In[1]:= SetDirectory[NotebookDirectory[]]
Out[1]= /Users/mauel/Library/CloudStorage/Dropbox/Work/Courses/Plasma I/Plasma 1 (2023)

In[2]:= boxLength = 500; (* Length of plasma and number of grid points *)
In[3]:= numElectrons = 4000; (* also number of ions *)
In[4]:= qOvere = 4.0*^-4; (* q_e/epsilon_0 << 1 *)
In[5]:= electronTemp = 0.25; (* kT_e; v_the = sqrt(kT_e) *)
In[6]:= Print["Average Density = ", nAvg = numElectrons / boxLength];
Print["Electron Thermal Velocity = ", vthe = Sqrt[electronTemp]];
Print["Plasma Frequency = ", wp = Sqrt[(numElectrons / boxLength) qOvere]];
Print["Debye Length = ", lambdaD = vthe / wp];
Print["Number of particles per cell = ", nAvg];
Print["Number of Debye lengths in box = ", boxLength / lambdaD];
Print["Number of time steps per plasma oscillation = ", 1 / wp];

Average Density = 8
Electron Thermal Velocity = 0.5
Plasma Frequency = 0.0565685
Debye Length = 8.83883
Number of particles per cell = 8
Number of Debye lengths in box = 56.5685
Number of time steps per plasma oscillation = 17.6777
```

Electrons


Each particle is represented by a coordinate, $\{x_i, v_i\}$.

```
In[13]:= electrons = Table[{RandomReal[{0, boxLength}],  
    RandomVariate[NormalDistribution[0.0, vthe]]}, {numElectrons}];
```


```
In[14]:= Dimensions[electrons]
```

```
Out[14]=  
{4000, 2}
```

```
In[15]:= ListPlot[electrons, AxesLabel → {"x", "v"}, PlotLabel → "Electron Phase Space"]
```



```
In[16]:= Plot[PDF[NormalDistribution[0, vthe], x], {x, -2, 2}, Filling → Axis];  
Histogram[electrons[[All, 2]], Automatic, "PDF"];  
pltElectDistribution = Show[%, %, AxesLabel → {"v", "F(v)"}]
```



The electron charge density is evaluated on the *grid points*. We use "triangular weighting" to map the electron's position (in-between the grid points) to the charge density at each point.

```
In[19]:= rho0 = Table[0.0, {boxLength+1}]; (* including 0...boxLength *)
rhoT = rho0;
```

```
In[21]:= addρ[x_Real] := Module[{loc},
  loc = Floor[x];
  rhoT[[loc+1]] += (loc+1-x);
  rhoT[[loc+2]] += (x-loc);]
```

```
In[22]:= getρ[e_List] := Module[{},
  rhoT = rho0;
  addρ /@ First[Transpose[e]];
  rhoT[[1]] += rhoT[[boxLength+1]];
  Drop[rhoT, -1]]
```


```
In[23]:= rhoElectron = getρ[electrons]; // Timing
```

```
Out[23]:= {0.03048, Null}
```


```
In[24]:= Plus @@ rhoElectron (* How many electrons? *)
```

```
Out[24]:= 4000.
```


```
In[25]:= pr1 = ListPlot[rhoElectron]
```



```
In[26]:= pr2 = ListPlot[MovingAverage[rhoElectron, Floor[λD]],  
    Joined → True, PlotStyle → {Red, Thickness[0.01]}]
```



```
In[27]:= Show[pr1, pr2, PlotLabel → "Electron Density",  
    AxesLabel → {"x", "ρ"}, PlotRange → {0, 2 nAvg}]
```



Notice large fluctuations in the electron density exists. Why? Answer: the amount of "noise" in our simulation depends upon the number of particles. On average, we have only a few particles for cell. The (random) statistical fluctuations scale as $\pm \sqrt{n}$, where n is the average number of particles per cell. For $n = 4$, we get $\pm 50\%$ noise!

Ions

We take the ion density to be a *constant* and equal to the average electron density.

```
In[28]:= ionDensity = numElectrons / (boxLength) // N
```


Out[28]=

8.

Finding the Electric Field


```
In[29]:= pr1 = ListPlot[ionDensity - rhoElectron]
```

Out[29]=




```
In[30]:= pr2 = ListPlot[MovingAverage[ionDensity - rhoElectron, Floor[ $\lambda$ D]],  
    Joined  $\rightarrow$  True, PlotStyle  $\rightarrow$  {Green, Thickness[0.01]}]
```

Out[30]=



```
In[31]:= Show[pr1, pr2, PlotLabel → "Charge Density", AxesLabel → {"x", "ρ"}]
Out[31]=
```




```
In[32]:= kFourier = Table[2 Sin[2.0 π n / (2 boxLength)], {n, 0, boxLength - 1}];
(* This represents the "effective" wavenumber
   used to invert Poisson's equation with the FFT. *)
```

```
In[33]:= kFourier2 = kFourier^2;
```

```
In[34]:= ListPlot[ $\frac{1}{\text{Drop}[kFourier2, 1]}$ , PlotRange → {0, 25},
  PlotLabel → "1/k2", AxesLabel → {"k", ""}, Joined → True]
```

```
Out[34]=
```




For a given electron and ion density, this computes the electric field using an FFT to invert Poisson's equation...

```
In[35]:= getE[rhoE_] := Module[{ϕ},
  ϕ = {0.0} ~Join~ (Drop[Fourier[ionDensity - rhoE], 1] / Drop[kFourier2, 1]);
  ϕ = Re[InverseFourier[qOverϕ]];
  - (RotateLeft[ϕ] - RotateRight[ϕ]) / 2]
```

```
In[36]:= eField = getE[rhoElectron]; // Timing
```

```
Out[36]:= {0.002253, Null}
```

```
In[37]:= ListPlot[eField, PlotLabel → "Electric Field", AxesLabel → {"x", "E"}]
```



Of course, with a "noisy" charge density, we also get a "noisy" electric field!

Electron and Ion Step

In this section, we use the "Euler-Cromer" method to advance the position and velocity of a particle for a given electric field.

```
In[38]:= stepE[eField_List][{x_, v_}] := Module[{loc, locp, force, vNew},
  loc = Floor[x];
  locp = Mod[loc + 1, boxLength] + 1;
  force = eField[[loc + 1]] (loc + 1 - x) + eField[[locp]] (x - loc);
  (* advance velocity *)
  vNew = v - force;
  {Mod[x + vNew, boxLength], vNew}];
```

```
In[39]:= (* 1st Time is slower *) (ParallelMap[stepE[eField][#] &, electrons];) // Timing
```

```
Out[39]:= {1.3489, Null}
```


```
In[40]:= (electrons1 = ParallelMap[stepE[eField][#] &, electrons];) // Timing
```

```
Out[40]:= {0.10458, Null}
```

View of the change in the particles' phase-space positions...


```
In[41]:= ListPlot[electrons1 - electrons, Frame → True]
```

```
Out[41]=
```



Many Steps

Now, we can repeat these calculations for many times steps. Be warned: as we increase the number of particles, this becomes a *very long* computation...

```
In[42]:= numSteps = 60;
```

```
(* How many plasma periods? *) (numSteps 5)  $\omega p$  / (2  $\pi$ )
```

```
Out[43]=
```

```
16.9706
```

```
In[61]:= pInd1 = 0; ProgressIndicator[Dynamic[pInd1], {1, numSteps}]
```

```
Timing[Do[
```

```
  Do[rhoElectron = get $\rho$ [electrons];
```

```
    eField = getE[rhoElectron];
```

```
    electrons = ParallelMap[stepE[eField][#1] &, electrons];, {5}];
```

```
pInd1 = i;
```

```
ePlot1[i] = GraphicsRow[{ListPlot[eField, PlotLabel → "Electric Field",
  AxesLabel → {"x", "E"}, PlotRange → {-0.01, 0.01}, Joined → True], ListPlot[
  electrons, AxesLabel → {"x", "v"}, PlotRange → {{1, boxLength}, {-1.5, 1.5}},
  PlotLabel → "Electron Phase Space"]}], {i, numSteps}];]
```

```
Out[61]=
```




```
Out[62]=
```

```
{123.036, Null}
```

```
In[63]:= AnimationVideo[Show[ePlot1[i]], {i, 1, numSteps, 1}]
Export["Efield-e1.mov", %, "QuickTime"]
```

Out[63]=




Out[64]=

Efield-e1.mov

```
In[65]:= pr1 = ListPlot[rhoElectron];
pr2 = ListPlot[MovingAverage[rhoElectron, Floor[λD]],
  Joined → True, PlotStyle → {Red, Thickness[0.01]}];
Show[pr1, pr2, PlotLabel → "Electron Density", AxesLabel → {"x", "ρ"}]
```

Out[67]=




The mean-squared average velocity of the electrons and ions...

```
In[68]:= (Plus @@ (Sqrt[#^2] & /@ Last[Transpose[electrons]])) / numElectrons
```

Out[68]=

0.401685

```
In[69]:= ListPlot[electrons, AxesLabel → {"x", "v"}, PlotLabel → "Electron Phase Space"]
Out[69]=
```




Electron Beam and Electrons

What happens with a different initial condition?


In this section, we examine counter-drifting electrons.

```
In[70]:= beamFraction = 0.2;
In[71]:= electrons =
  Table[{RandomReal[{0, boxLength}], RandomVariate[NormalDistribution[0.0, vthe]]},
    {numElectrons - Floor[beamFraction numElectrons]}] ~Join~
  Table[{RandomReal[{0, boxLength}], 3 vthe (1 + RandomReal[{-0.1, 0.1}])},
    {Floor[beamFraction numElectrons]}];
In[72]:= Plot[PDF[NormalDistribution[0, vthe], x], {x, -2, 2}, Filling → Axis];
Histogram[electrons[[All, 2]], Automatic, "PDF"];
Show[%, %, AxesLabel → {"v", "F(v)"}]
```



```
In[75]:= ListPlot[electrons, AxesLabel → {"x", "v"}, PlotLabel → "Election Phase Space"]
```

```
Out[75]=
```



Many Steps

```
In[76]:= numSteps = 150;
```

```
(* How many plasma periods? *) (numSteps 5)  $\omega_p$  / (2  $\pi$ )
```

```
Out[77]=
```

```
42.4264
```

```
In[78]:= eFieldSave = {};
```

```
pInd2 = 0; ProgressIndicator[Dynamic[pInd2], {1, numSteps}]
```

```
Timing[Do[
```

```
  Do[rhoElectron = get $\rho$ [electrons];
```

```
    eField = getE[rhoElectron];
```

```
    electrons = ParallelMap[stepE[eField][#1] &, electrons];
```

```
    , {5}];
```

```
eFieldSave = Append[eFieldSave, eField];
```

```
pInd2 = i;
```

```
ePlot2[i] = GraphicsRow[{ListPlot[eField, PlotLabel → "Electric Field",
```

```
  AxesLabel → {"x", "E"}, PlotRange → {-0.02, 0.02}, Joined → True],
```

```
ListPlot[electrons, AxesLabel → {"x", "v"}, PlotLabel →
```

```
  "Electron Phase Space", PlotRange → {-1.5, 2.0}]]], {i, numSteps}];]
```

```
Out[79]=
```




```
Out[80]=
```

```
{305.939, Null}
```

```
In[81]:= AnimationVideo[Show[ePlot2[i]], {i, 1, numSteps, 1}]
Export["Efield-e2.mov", %, "QuickTime"]
```

Out[81]=




Out[82]=

Efield-e2.mov


```
In[83]:= ListPlot[electrons, AxesLabel → {"x", "v"}, PlotLabel → "Electron Phase Space"]
```

Out[83]=



```
In[84]:= ListPlot[eField, PlotLabel → "Electric Field", AxesLabel → {"x", "E"}, Joined → True]
```

```
Out[84]=
```




```
In[85]:= Dimensions[eFieldSave]
```

```
Out[85]=
```

```
{150, 500}
```

```
In[86]:= ListDensityPlot[Transpose[eFieldSave],  
  AspectRatio → 1 / GoldenRatio, FrameLabel → {"time", "Length"}]
```

```
Out[86]=
```



```
In[87]:= Dimensions[Fourier[eFieldSave]]  
Dimensions[Abs[Fourier[eFieldSave][[1 ;; numSteps / 2, 1 ;; boxLength / 2]]]  
Max[Abs[Fourier[eFieldSave][[1 ;; numSteps / 2, 1 ;; boxLength / 2]]]
```

```
Out[87]=
```

```
{150, 500}
```


```
Out[88]=
```

```
{75, 250}
```


```
Out[89]=
```

```
0.171019
```

```
In[90]:= ListDensityPlot[
  Abs[Fourier[eFieldSave]][[1 ;; Floor[numSteps / 4], 1 ;; Floor[boxLength / 10]]],
  AspectRatio → 1 / GoldenRatio, FrameLabel → {"k", " $\omega$ "}, PlotRange → {0, 0.2}]
```



```
In[91]:= pr1 = ListPlot[rhoElectron];
pr2 = ListPlot[MovingAverage[rhoElectron, Floor[λD]],
  Joined → True, PlotStyle → {RGBColor[1, 0, 0], Thickness[0.01]}];
Show[pr1, pr2, PlotLabel → "Electron Density", AxesLabel → {"x", "ρ"}]
```



Counter-Drifting Electrons

What happens with a different initial condition?


In this section, we examine counter-drifting electrons.

```
In[123]:= electrons = Table[{RandomReal[{0, boxLength}], (RandomInteger[{0, 1}] - 0.5)
  vthe 4 (1 + RandomReal[{-0.1, 0.1}])}], {numElectrons}];
```

In[124]:=

```
Plot[PDF[NormalDistribution[0, vthe], x], {x, -2, 2}, Filling → Axis];
Histogram[electrons[[All, 2]], Automatic, "PDF"];
pltInitialElectronDist = Show[%, %, AxesLabel → {"v", "F(v)"}]
```


Out[126]=



In[127]:=

```
ListPlot[electrons, AxesLabel → {"x", "v"}, PlotLabel → "Election Phase Space"]
```

Out[127]=



Many Steps

In[128]:=

```
numSteps = 150;
```

In[129]:=

```
(* How many plasma periods? *) (numSteps 5)  $\omega_p$  / (2  $\pi$ )
```

Out[129]=

```
6.75237
```

In[130]:=

```
eFieldSave = {};
```


In[131]:=

```

pInd3 = 0; ProgressIndicator[Dynamic[pInd3], {1, numSteps}]
Timing[Do[
  Do[rhoElectron = getρ[electrons];
    eField = getE[rhoElectron];
    electrons = ParallelMap[stepE[eField][#1] &, electrons];
    , {5}];
  eFieldSave = Append[eFieldSave, eField];
  pInd3 = i;
  ePlot3[i] = GraphicsRow[{ListPlot[eField, PlotLabel → "Electric Field",
    AxesLabel → {"x", "E"}, PlotRange → {-0.04, 0.04}, Joined → True],
    ListPlot[electrons, AxesLabel → {"x", "v"}, PlotLabel →
      "Electron Phase Space", PlotRange → {-2.0, 2.0}]}], {i, numSteps}];]

```

Out[131]=



Out[132]=

```
{305.067, Null}
```


In[133]:=

```

AnimationVideo[Show[ePlot3[i]], {i, 1, numSteps, 1}]
Export["Efield-e3.mov", %, "QuickTime"]

```

Out[133]=




Out[134]=

```
Efield-e3.mov
```

In[135]:=

```
ListPlot[electrons, AxesLabel → {"x", "v"}, PlotLabel → "Electron Phase Space"]
```


Out[135]=



In[146]:=

```
Plot[PDF[NormalDistribution[0, vthe], x], {x, -2, 2}, Filling → Axis];
Histogram[electrons[[All, 2]], Automatic, "PDF", ChartStyle → Green];
Show[%, pltInitialElectronDist, AxesLabel → {"v", "F(v)"}]
```


Out[148]=



In[139]:=

```
ListPlot[eField, PlotLabel → "Electric Field", AxesLabel → {"x", "E"}]
```


Out[139]=



In[140]:=

```
ListDensityPlot[Transpose[eFieldSave],  
  AspectRatio → 1 / GoldenRatio, FrameLabel → {"time", "Length"}]
```

Out[140]=



In[141]:=

```
Max[Abs[Fourier[eFieldSave][[1 ;; numSteps / 2, 1 ;; boxLength / 2]]]
```


Out[141]=

2.01855

In[142]:=

```
ListDensityPlot[
  Abs[Fourier[eFieldSave][[1 ;; Floor[numSteps / 4], 1 ;; Floor[boxLength / 10]]],
  AspectRatio → 1 / GoldenRatio, FrameLabel → {"k", " $\omega$ "}, PlotRange → {0, 0.5}]
```


Out[142]=



In[143]:=

```
pr1 = ListPlot[rhoElectron];
pr2 = ListPlot[MovingAverage[rhoElectron, Floor[λD]],
  Joined → True, PlotStyle → {RGBColor[1, 0, 0], Thickness[0.01]}];
Show[pr1, pr2, PlotLabel → "Electron Density", AxesLabel → {"x", "ρ"}]
```

Out[145]=



Summary

A one-dimensional "particle-in-cell" simulation of a plasma was defined using *Mathematica*. PIC simulations represent a direct model of a plasma allowing computation of dynamical and statistical properties of plasmas. We examined three particularly easy examples determined by the initial conditions of the distributions of ions and electrons. (Many more examples are possible—even with this simple model!) In one example, we looked at "noise" in a *stable* plasma. In the second example, we looked at

the evolution of counter-streaming electrons that excite the so-called "two-stream" instability. The initial kinetic energy of the electrons excites growing electrostatic oscillations. Particle energy is converted into electric-field energy. In the third example, we looked at a "clump" of plasma and observed "ambipolar" electron confinement and the plasma Debye length.